
/*********************************************************************************
** 
/** SCRIPT:  LR_200_Class.txt 
/** CREATED BY: RENEE SCHICKER 
/** DATE CREATED: 08 DECEMBER 2009 
/** DATE MODIFIED: 
/** 
/** SCRIPTS USED: Checkproggrid.txt D:\renee_gis\scripts\checkproggrid.txt 
/**  
/** GRIDS USED:  
/**        MyBnds  D:\Renee_gis\output_data\Organised\03_DEM\MyBnds 
/**        logiteq2int  
/**  D:\Renee_GIS\Output_data\Organised\15_Multi_Reg\200m_set1\logiteq2int  
/** 
/** 
/** COVERS MADE: Set1NBCov Set1NBLICov Set1EICov Set1EILICov  
/**   Set1GICov Set1GILICov Set1QCov Set1QLICov  
/**   Set1DICov Set1DILICov  
/** 
/** PURPOSE: Classify the output of the logistic regression result 
/**   (using the integer grid version of set 1 at 200 m) using 
/**   five different classification techniques available in ArcMap.  
/**   Natural Breaks (Jenks), Equal Interval, Geometric Interval, 
/**   Quantile, and Defined Interval (set at 20%) have been selected 
/**   for analysis. Once classified grids have been created these are  
/**   converted to cover then unioned with the landslide inventory. 
/**   The unioned layer can then be submitted to zonal statistic in  
/**   ArcMap to determine the area for each combination of susceptibility 
/**   class and presence or absence of landslides.  
/**   The conversion of the grid in continuous integer format to cover 
/**   does not always work (sometimes too much information to handle) as 
/**   pixel size is 25 m and there is over 38 million in the Waikato region 
/**   taking on values between 1 and 100.  
/**    
/** NOTE:  The workspace is set at the start of each of the two primary routines 
/**   So for set 1 all files will be saved to: 
/**   D:\Renee_GIS\Output_data\Organised\15_Multi_Reg\200m_set1 
/** 
/** WHAT TO DO: Use the break values obtained from applying the different  
/**   classifications to the continuous integer grid output in ArcMap 
/**   in the section below (SET CLASS BREAKS AS VARIABLES).  
/**   Set the PROCESSES TO CALL to run set 1. 
/** 
/**---------------------------------- HISTORY -------------------------------------- 
/** 08 DECEMBER 2009 Script created and explained 
/********************************************************************************* 
 
 
/********************************************************************************* 
/**   SET CLASS BREAKS AS VARIABLES  
/********************************************************************************* 



/* Four breaks are used to determine the five classes VL (1), L (2), M (3), H (4), VH (5) 
/* which are equivalent to Very Low, Low, Moderate, High, Very High. 
/* Breaks are in ascending order (Lowest to highest 1 - 4) 
/* If ArcMap gave breaks with decimals for an integer grid, round down. 
/* Defined Intervals have been set in intervals of 20% later in the script. 
/*  
/* Understanding the variables:  
/* 
/* S1 = Set 1 NB = Natural Breaks  B1 = Break 1 
/*   EI = Equal Interval  B2 = Break 2 
/*   GI = Geometric Interval  B3 = Break 3 
/*    Q = Quantile    B4 = Break 4 
/**   DI = Defined Interval 
/* 
/*~~~~~~~~~~~~~~~~~~~~~~ Natural Breaks (NB) ~~~~~~~~~~~~~~~~~~~~~~ 
/* SET 1: 
&SETVAR .S1_NB_B1 = 27 
&SETVAR .S1_NB_B2 = 40 
&SETVAR .S1_NB_B3 = 55 
&SETVAR .S1_NB_B4 = 72 
 
/*~~~~~~~~~~~~~~~~~~~~~~ Equal Interval (EI) ~~~~~~~~~~~~~~~~~~~~~~ 
/* SET 1: 
&SETVAR .S1_EI_B1 = 27 
&SETVAR .S1_EI_B2 = 45 
&SETVAR .S1_EI_B3 = 63 
&SETVAR .S1_EI_B4 = 81 
 
/*~~~~~~~~~~~~~~~~~~~~~~ Geometric Interval (GI) ~~~~~~~~~~~~~~~~~~~~~~ 
/* SET 1: 
&SETVAR .S1_GI_B1 = 22 
&SETVAR .S1_GI_B2 = 30 
&SETVAR .S1_GI_B3 = 43 
&SETVAR .S1_GI_B4 = 64 
 
/*~~~~~~~~~~~~~~~~~~~~~~ Quantile (Q) ~~~~~~~~~~~~~~~~~~~~~~ 
/* SET 1: 
&SETVAR .S1_Q_B1 = 25 
&SETVAR .S1_Q_B2 = 32 
&SETVAR .S1_Q_B3 = 43 
&SETVAR .S1_Q_B4 = 60 
 
 
/********************************************************************************* 
/**    PROCESSES TO CALL 
/********************************************************************************* 
 
&CALL Set_1 
&CALL OutputSet1 
 
&RETURN 



 
/********************************************************************************* 
/**    PRIMARY ROUTINES 
/********************************************************************************* 
&ROUTINE Set_1 
 
Workspace D:\Renee_GIS\Output_data\Organised\15_Multi_Reg\200m_set1 
 
&TYPE SET 1 NATURAL BREAKS 
&TYPE 
&SETVAR .B1 = %.S1_NB_B1% 
&SETVAR .B2 = %.S1_NB_B2% 
&SETVAR .B3 = %.S1_NB_B3% 
&SETVAR .B4 = %.S1_NB_B4% 
&CALL Classify 
&IF [EXIST Set1NBCov -COVER] &THEN KILL Set1NBCov  
&IF [EXIST Set1NBLICov -COVER] &THEN KILL Set1NBLICov 
GRIDPOLY LSIClass Set1NBCov  
KILL LSIClass  
UNION D:\Renee_GIS\Output_data\Organised\04_Inventory\li_extent Set1NBCov Set1NBLICov 
 
&TYPE SET 1 EQUAL INTERVAL 
&TYPE 
&SETVAR .B1 = %.S1_EI_B1% 
&SETVAR .B2 = %.S1_EI_B2% 
&SETVAR .B3 = %.S1_EI_B3% 
&SETVAR .B4 = %.S1_EI_B4% 
&CALL Classify 
&IF [EXIST Set1EICov -COVER] &THEN KILL Set1EICov  
&IF [EXIST Set1EILICov -COVER] &THEN KILL Set1EILICov 
 
GRIDPOLY LSIClass Set1EICov  
KILL LSIClass  
UNION D:\Renee_GIS\Output_data\Organised\04_Inventory\li_extent Set1EICov Set1EILICov 
 
&TYPE SET 1 GEOMETRIC INTERVAL 
&TYPE 
&SETVAR .B1 = %.S1_GI_B1% 
&SETVAR .B2 = %.S1_GI_B2% 
&SETVAR .B3 = %.S1_GI_B3% 
&SETVAR .B4 = %.S1_GI_B4% 
&CALL Classify 
&IF [EXIST Set1GICov -COVER] &THEN KILL Set1GICov  
&IF [EXIST Set1GILICov -COVER] &THEN KILL Set1GILICov 
GRIDPOLY LSIClass Set1GICov  
KILL LSIClass  
UNION D:\Renee_GIS\Output_data\Organised\04_Inventory\li_extent Set1GICov Set1GILICov 
 
&TYPE SET 1 QUANTILE 
&TYPE 
/* SET breaks here (in ascending order) 



&SETVAR .B1 = %.S1_Q_B1% 
&SETVAR .B2 = %.S1_Q_B2% 
&SETVAR .B3 = %.S1_Q_B3% 
&SETVAR .B4 = %.S1_Q_B4% 
&CALL Classify 
&IF [EXIST Set1QCov -COVER] &THEN KILL Set1QCov  
&IF [EXIST Set1QLICov -COVER] &THEN KILL Set1QLICov 
GRIDPOLY LSIClass Set1QCov  
KILL LSIClass  
UNION D:\Renee_GIS\Output_data\Organised\04_Inventory\li_extent Set1QCov Set1QLICov 
 
&TYPE SET 1 DEFINED INTERVAL: 20 percent 
&TYPE 
/* SET breaks here (in ascending order) 
&SETVAR .B1 = 20 
&SETVAR .B2 = 40 
&SETVAR .B3 = 60 
&SETVAR .B4 = 80 
&CALL Classify 
&IF [EXIST Set1DICov -COVER] &THEN KILL Set1DICov  
&IF [EXIST Set1DILICov -COVER] &THEN KILL Set1DILICov 
GRIDPOLY LSIClass Set1DICov  
KILL LSIClass  
UNION D:\Renee_GIS\Output_data\Organised\04_Inventory\li_extent Set1DICov Set1DILICov 
&RETURN 
 
/*~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~ 
&ROUTINE OutputSet1 
Workspace D:\Renee_GIS\Output_data\Organised\15_Multi_Reg\200m_set1 
 
&TYPE SET 1 NATURAL BREAKS 
&TYPE 
&SETVAR .B1 = %.S1_NB_B1% 
&SETVAR .B2 = %.S1_NB_B2% 
&SETVAR .B3 = %.S1_NB_B3% 
&SETVAR .B4 = %.S1_NB_B4% 
&CALL Classify 
&RETURN 
 
 
/********************************************************************************* 
/**    SECONDARY ROUTINE 
/********************************************************************************* 
&ROUTINE Classify 
&CALL Setup_Grid 
VL = CON(logiteq2int le %.B1%, 1, 0) 
L = CON(logiteq2int gt %.B1% and logiteq2int le %.B2%, 2, 0) 
M = CON(logiteq2int gt %.B2% and logiteq2int le %.B3%, 3, 0) 
H = CON(logiteq2int gt %.B3% and logiteq2int le %.B4%, 4, 0) 
VH = CON(logiteq2int gt %.B4%, 5, 0) 
LSIClass = (VL + L + M + H + VH) 



KILL VL 
KILL L 
KILL M 
KILL H 
KILL VH 
SETMASK off 
Q 
&RETURN 
/********************************************************************************* 
/**    TERTIARY ROUTINE 
/********************************************************************************* 
&ROUTINE Setup_Grid 
&RUN D:\renee_gis\scripts\checkproggrid.txt 
&TYPE set window to MyBnds... 
SETWINDOW D:\Renee_gis\output_data\Organised\03_DEM\MyBnds 
&TYPE Set mask to MyBnds... 
SETMASK D:\Renee_gis\output_data\Organised\03_DEM\MyBnds 
&TYPE Mask Set 
&RETURN 


